Experimenting with Carmo and Jones' DDL

Lavanya Singh

October 29, 2021

Contents

1	System Definition					
	1.1	Definitions	3			
	1.2	Axiomatization	3			
	1.3	Abbreviations	4			
	1.4	Consistency	5			
2	Inference Rules					
	2.1	Basic Inference Rules	7			
	2.2	Fancier Inference Rules	7			
3	Axioms					
	3.1	Box	8			
	3.2	0	8			
	3.3	Possible Box	9			
	3.4	Actual Box	9			
	3.5	Relations Between the Modal Operators	10			
4	The Categorical Imperative					
	4.1	Simple Formulation of the Kingdom of Ends	11			
5	The	e Categorical Imperative	13			
	5.1	Simple Formulation of the Formula of Universal Law	13			
	5.2	Basic Tests	13			
	5.3	Specifying the Model	14			
	5.4	Consistent Sentences	15			
	5.5	Metaethical Tests	16			
6	Kroy's Formalization of the Categorical Imperative					
	6.1	The Substitution Operator	18			
		6.1.1 Definition	18			
		6.1.2 Abbreviations	18			
	6.2	Differences Between Kroy's Logic (Kr) and DDL	19			

6.3	The C	Categorical Imperative		20
6.4	Tests			21
	6.4.1	Specifying the Model	. 4	21
	6.4.2	Metaethical Tests	. 4	21
	6.4.3	Krov's Tests		21

```
theory carmojones-DDL
imports
Main
```

Referencing Benzmueller, Farjami, and Parent's implementation [1]

This theory contains the axiomatization of the system and some useful abbreviations.

1 System Definition

1.1 Definitions

This section contains definitions and constants necessary to construct a DDL model.

typedecl i — i is the type for a set of possible worlds."

```
type-synonym t = (i \Rightarrow bool)
```

- t represents a set of DDL formulas.
- this set is defined by its truth function, mapping the set of worlds to the formula set's truth value.

```
— accessibility relations map a set of worlds to:
```

consts $av::i \Rightarrow t$ — actual versions of that world set

- these worlds represent what is "open to the agent"
- for example, the agent eating pizza or pasta for dinner might constitute two different actual worlds

```
consts pv::i \Rightarrow t — possible versions of that world set
```

- these worlds represent was was "potentially open to the agent"
- for example, what someone across the world eats for dinner might constitute a possible world, since the agent has no control over this

```
consts ob::t \Rightarrow (t \Rightarrow bool) — set of propositions obligatory in this "context" — ob(context)(term) is True if t is obligatory in the context
```

 $\mathbf{consts}\ cw{::}i\ -$ current world

1.2 Axiomatization

This subsection contains axioms. Because the embedding is semantic, these are just constraints on models.

This axiomatization comes from [2] p6 and the HOL embedding defined in Benzmuller and Parent

```
axiomatization where
ax-3a: \forall w. \exists x. av(w)(x)
— every world has some actual version
and ax-4a: \forall w \ x. \ av(w)(x) \longrightarrow pv(w)(x)
— all actual versions of a world are also possible versions of it
and ax-4b: \forall w. pv(w)(w)
 — every world is a possible version of itself
and ax\text{-}5a: \forall X.\neg ob(X)(\lambda w. False)
— in any arbitrary context X, something will be obligatory
and ax-5b: \forall X Y Z. (\forall w. ((X(w) \land Y(w)) \longleftrightarrow (X(w) \land Z(w)))) \longrightarrow (ob(X)(Y))
\longleftrightarrow ob(X)(Z) — note that X(w) denotes w is a member of X
— X, Y, and Z are sets of formulas
— If X \cap Y = X \cap Z then the context X obliges Y iff it obliges Z
— ob(X)(\lambda \text{ w. Fw}) can be read as F \in ob(X)
and ax-5c2: \forall X Y Z. (((\exists w. (X(w) \land Y(w) \land Z(w))) \land ob(X)(Y) \land ob(X)(Z)))
\longrightarrow ob(X)(\lambda w. Y(w) \wedge Z(w))
and ax-5d: \forall X \ Y \ Z. \ ((\forall w. \ Y(w) \longrightarrow X(w)) \land ob(X)(Y) \land (\forall w. \ X(w) \longrightarrow Z(w)))
  \longrightarrow ob(Z)(\lambda w.(Z(w) \land \neg X(w)) \lor Y(w))
— If some subset Y of X is in ob(X) then in a larger context Z, any obligatory
proposition must either be in Y or in Z-X
and ax-5e: \forall X \ Y \ Z. ((\forall w. \ Y(w) \longrightarrow X(w)) \land ob(X)(Z) \land (\exists w. \ Y(w) \land Z(w)))
\longrightarrow ob(Y)(Z)
```

1.3 Abbreviations

Z shares some elements with Y

These abbreviations are defined in @citeBenzmullerParent p9

These are all syntactic sugar for HOL expressions, so evaluating these symbols will be light-weight

— If Z is obligatory in context X, then Z is obligatory in a subset of X called Y, if

```
— propositional logic symbols abbreviation ddlneg::t\Rightarrow t \ (\neg) where \neg A \equiv \lambda w. \ \neg A(w) abbreviation ddlor::t\Rightarrow t\Rightarrow t \ (-\lor-) where A\lor B \equiv \lambda w. \ (A(w)\lor B(w)) abbreviation ddland::t\Rightarrow t\Rightarrow t \ (-\land-) where A\land B \equiv \lambda w. \ (A(w)\land B(w)) abbreviation ddlif::t\Rightarrow t\Rightarrow t \ (-\to-) where A\rightarrow B \equiv (\lambda w. \ A(w) \longrightarrow B(w))
```

```
abbreviation ddlequiv::t\Rightarrow t\Rightarrow t (-\equiv -)
  where (A \equiv B) \equiv ((A \rightarrow B) \land (B \rightarrow A))
 — modal operators
abbreviation ddlbox::t \Rightarrow t (\square)
  where \Box A \equiv \lambda w. \forall y. A(y)
abbreviation ddldiamond::t \Rightarrow t \ (\lozenge)
  where \Diamond A \equiv \neg(\Box(\neg A))
— O\{B|A\} can be read as "B is obligatory in the context A"
abbreviation ddlob::t\Rightarrow t\Rightarrow t (O\{-|-\})
  where O\{B|A\} \equiv \lambda \ w. \ ob(A)(B)
— modal symbols over the actual and possible worlds relations
abbreviation ddlboxa::t\Rightarrow t (\square_a)
  where \Box_a A \equiv \lambda x. \forall y. (\neg av(x)(y) \lor A(y))
abbreviation ddldiamonda::t \Rightarrow t \ (\lozenge_a)
  where \lozenge_a A \equiv \neg(\Box_a(\neg A))
abbreviation ddlboxp::t\Rightarrow t (\square_p)
  where \Box_p A \equiv \lambda x. \forall y. (\neg pv(x)(y) \lor A(y))
abbreviation ddldiamondp::t \Rightarrow t \ (\lozenge_p)
  where \Diamond_p A \equiv \neg(\Box_a(\neg A))
— obligation symbols over the actual and possible worlds
abbreviation ddloba::t \Rightarrow t (O_a)
  where O_a A \equiv \lambda x. ob(av(x))(A) \wedge (\exists y.(av(x)(y) \wedge \neg A(y)))
abbreviation ddlobp::t \Rightarrow t (O_p)
  where O_p A \equiv \lambda x. \ ob(pv(x))(A) \wedge (\exists y.(pv(x)(y) \wedge \neg A(y)))
— syntactic sugar for a "monadic" obligation operator
abbreviation ddltrue::t(\top)
  where \top \equiv \lambda w. True
abbreviation ddlob-normal::t \Rightarrow t (O \{-\})
  where (O\{A\}) \equiv (O\{A|\top\})
— validity
abbreviation ddlvalid::t\Rightarrow bool (\models-)
  where \models A \equiv \forall w. \ A \ w
abbreviation ddlvalidcw::t \Rightarrow bool\ (\models_c-)
  where \models_c A \equiv A \ cw
```

1.4 Consistency

Consistency is so easy to show in Isabelle!

```
\mathbf{lemma} \ \mathit{True} \ \mathbf{nitpick} \ [\mathit{satisfy}, \mathit{user-axioms}, \mathit{show-all}, \mathit{format} = 2] \ \mathbf{oops}
```

- Nitpick successfully found a countermodel.
- It's not shown in the document printout, hence the oops.
- If you hover over "nitpick" in JEdit, the model will be printed to output.

 \mathbf{end}

theory carmojones-DDL-completeness imports carmojones-DDL

begin

This theory shows completeness for this logic with respect to the models presented in carmojonesDDl.thy.

2 Inference Rules

2.1 Basic Inference Rules

These inference rules are common to most modal and propostional logics

```
lemma modus-ponens: assumes \models A assumes \models (A \rightarrow B) shows \models B using assms(1) assms(2) by blast— Because I have not defined a "derivable" operator, inference rules are written using assumptions.
```

— For further meta-logical work, defining metalogical operators may be useful

```
lemma nec: assumes \models A shows \models (\Box A) by (simp \ add: assms)
lemma nec-a: assumes \models A shows \models (\Box_a A) by (simp \ add: assms)
lemma nec-p: assumes \models A shows \models (\Box_p A) by (simp \ add: assms)
```

2.2 Fancier Inference Rules

These are new rules that Carmo and Jones introduced for this logic.

```
\begin{array}{l} \mathbf{lemma} \ \textit{Oa-boxaO}\colon\\ \mathbf{assumes} \models (B \rightarrow ((\neg(\Box((O_a \ A) \rightarrow ((\Box_a w) \land \ O\{A|w\}))))))\\ \mathbf{shows} \models (B \rightarrow (\neg(\Diamond(O_a \ A))))\\ \mathbf{oops}\\ \mathbf{lemma} \ \textit{Oa-boxpO}\colon\\ \mathbf{assumes} \models (B \rightarrow ((\neg(\Box((O_p \ A) \rightarrow ((\Box_p w) \land \ O\{A|w\}))))))\\ \mathbf{shows} \models (B \rightarrow (\neg(\Diamond(O_p \ A))))\\ \mathbf{oops} \end{array}
```

— The oops indicates that we were not able to find a proof for these lemmas.

B and A must not contain w. not sure how to encode that requirement. one option is to define a new free variables predicate and use that, but that requires a deeper embedding than I have. May become a problem later

3 Axioms

3.1 Box

lemma *O-nec*:

shows $\models (O\{B|A\} \rightarrow (\Box O\{B|A\}))$

```
— \square is an S5 modal operator, which is where these axioms come from.
lemma K:
 shows \models ((\Box(A \rightarrow B)) \rightarrow ((\Box A) \rightarrow (\Box B)))
 by blast
lemma T:
 \mathbf{shows} \models ((\Box A) \rightarrow A)
 by blast
lemma 5:
  \mathbf{shows} \models ((\Diamond A) \to (\Box(\Diamond A)))
 by blast
3.2
        0
This characterization of O comes from Carmo and Jones p 593
lemma O-diamond:
  \mathbf{shows} \models (O\{A|B\} \rightarrow (\Diamond(B \land A)))
  using ax-5b ax-5a
 by metis
— A is only obligatory in a context if it can possibly be true in that context.
lemma O-C:
 shows \models (((\Diamond (A \land (B \land C))) \land (O\{B|A\} \land O\{C|A\})) \rightarrow (O\{B \land C|A\}))
 by (metis ax-5c2)
— The conjunction of obligations in a context is obligatory in that context.
— The restriction \Diamond(ABC) is to prevent contradictory obligations and contexts.
lemma O-SA:
  \mathbf{shows} \models (((\Box(A \rightarrow B)) \land ((\Diamond(A \land C)) \land O\{C|B\})) \rightarrow (O\{C|A\}))
  using ax-5e by blast
— The principle of strengthening the antecedent.
lemma O-REA:
  shows \models ((\Box(A \equiv B)) \rightarrow (O\{C|A\} \equiv O\{C|B\}))
 using O-diamond ax-5e by blast

    Equivalence for equivalent contexts.

lemma O-contextual-REA:
 \mathbf{shows} \models ((\Box(C \to (A \equiv B))) \to (O\{A|C\} \equiv O\{B|C\}))
 by (metis ax-5b)
— The above lemma, but in some context C.
```

```
by simp
— Obligations are necessarily obligated.
lemma ax-5b'':
   shows ob X Y \longleftrightarrow ob X (\lambda z. (Y z) \land (X z))
   by (metis (no-types, lifting) ax-5b)
lemma O-to-O:
   shows \models (O\{B|A\} \rightarrow O\{(A \rightarrow B)|\top\})
proof-
   have \forall X \ Y \ Z. \ (ob \ X \ Y \ \land \ (\forall \ w. \ X \ w \longrightarrow Z \ w)) \longrightarrow ob \ Z \ (\lambda w.(Z \ w \ \land \neg X \ w) \ \lor
 Y(w)
   by (smt ax-5d ax-5b ax-5b'')
   thus ?thesis
   proof -
        have f1: \forall p \ pa \ pb. \ ((\neg (ob \ p \ pa)) \lor (\exists i. \ (p \land (\neg pb)) \ i)) \lor (ob \ pb \ ((pb \land (\neg pb)))) )
p))\vee pa))
         using \forall X \ Y \ Z. \ ob \ X \ Y \ \land (\models(X \rightarrow Z)) \longrightarrow ob \ Z \ (\ (Z \land (\neg X)) \lor \ Y) \land \mathbf{by} \ force
      obtain ii :: (i \Rightarrow bool) \Rightarrow (i \Rightarrow bool) \Rightarrow i where
          \forall x\theta \ x2. \ (\exists v3. \ (x2\land (\neg x\theta)) \ v3) = (x2\land (\neg x\theta)) \ (ii \ x\theta \ x2)
          by moura
       then have \forall p \ pa \ pb. ((\neg \ ob \ p \ pa) \lor (p \land (\neg \ pb)) \ (ii \ pb \ p)) \lor ob \ pb \ ((pb \land (\neg \ pb))) \ (ii \ pb \ p)) \lor ob \ pb \ ((pb \land (\neg \ pb))) \ (ii \ pb \ p)) \lor ob \ pb \ ((pb \land (\neg \ pb))) \ (ii \ pb \ p)) \lor ob \ pb \ ((pb \land (\neg \ pb))) \ (ii \ pb \ p)) \lor ob \ pb \ ((pb \land (\neg \ pb))) \ (ii \ pb \ p)) \lor ob \ pb \ ((pb \land (\neg \ pb))) \ (ii \ pb \ p)) \lor ob \ pb \ ((pb \land (\neg \ pb))) \ (ii \ pb \ p)) \lor ob \ pb \ ((pb \land (\neg \ pb))) \ (ii \ pb \ p)) \lor ob \ pb \ ((pb \land (\neg \ pb))) \ (ii \ pb \ p)) \lor ob \ pb \ ((pb \land (\neg \ pb))) \ (ii \ pb \ p)) \lor ob \ pb \ ((pb \land (\neg \ pb))) \ (ii \ pb \ p)) \lor ob \ pb \ ((pb \land (\neg \ pb))) \ (ii \ pb \ p)) \lor ob \ pb \ ((pb \land (\neg \ pb))) \ (ii \ pb \ p)) \lor ob \ pb \ ((pb \land (\neg \ pb))) \ (ii \ pb \ p)) \lor ob \ pb \ ((pb \land (\neg \ pb))) \ (ii \ pb \ p)) \lor ob \ pb \ ((pb \land (\neg \ pb))) \ (ii \ pb \ p)) \lor ob \ pb \ ((pb \land (\neg \ pb))) \ (ii \ pb \ p))
p))\vee pa)
          using f1 by presburger
      then show ?thesis
          by fastforce
   qed
qed
   - Moving from the dyadic to monadic obligation operators.
3.3
              Possible Box
 — \square_p is a KT modal operator.
lemma K-boxp:
   shows \models ((\Box_p(A \to B)) \to ((\Box_p A) \to (\Box_p B)))
   by blast
lemma T-boxp:
   shows \models ((\Box_n A) \to A)
   using ax-4b by blast
3.4
              Actual Box
— \square_a is a KD modal operator.
lemma K-boxa:
   shows \models ((\Box_a(A \to B)) \to ((\Box_a A) \to (\Box_a B)))
   by blast
lemma D-boxa:
   shows \models ((\Box_a A) \to (\Diamond_a A))
   using ax-3a by blast
```

3.5 Relations Between the Modal Operators

```
— Relation between \square, \square_a, and \square_p.
lemma box-boxp:
  shows \models ((\Box A) \to (\Box_p A))
  by auto
lemma boxp-boxa:
  shows \models ((\Box_p A) \to (\Box_a A))
  using ax-4a by blast
— Relation between actual and possible O and \square.
lemma not-Oa:
  shows \models ((\Box_a A) \rightarrow ((\neg(O_a \ A)) \land (\neg(O_a \ (\neg A)))))
  using O-diamond by blast
lemma not-Op:
\mathbf{shows} \models ((\Box_p A) \rightarrow ((\neg(O_p\ A)) \land (\neg(O_p\ (\neg A)))))
  using O-diamond by blast
lemma equiv-Oa:
  shows \models ((\Box_a (A \equiv B)) \rightarrow ((O_a \ A) \equiv (O_a \ B)))
  using O-contextual-REA by blast
lemma equiv-Op:
  shows \models ((\Box_p(A \equiv B)) \rightarrow ((O_p A) \equiv (O_p B)))
  using O-contextual-REA by blast
— relationships between actual and possible O and \square and O proper.
lemma factual-detach-a:
  shows \models (((O\{B|A\} \land (\Box_a A)) \land ((\Diamond_a B) \land (\Diamond_a (\neg B)))) \rightarrow (O_a B))
  using O-SA by auto
\mathbf{lemma}\ \mathit{factual}\text{-}\mathit{detach}\text{-}\mathit{p}\text{:}
  shows \models (((O\{B|A\} \land (\square_p A)) \land ((\lozenge_p B) \land (\lozenge_p (\neg B)))) \rightarrow (O_p B))
  by (smt O-SA boxp-boxa)
```

end

4 The Categorical Imperative

4.1 Simple Formulation of the Kingdom of Ends

This is my first attempt at formalizing the concept of the Kingdom of Ends

NOTE: this attempt revealed a bug in my embedding. I've included it as an artifact, but none of these theorems hold anymore (hence the oops).

```
abbreviation ddlpermissable::t \Rightarrow t \ (P-)
where (P \ A) \equiv (\neg(O \ \{\neg A\}))
```

- This operator represents permissibility
- Will be useful when discussing the categorical imperative
- Something is permissible if it is not prohibited
- Something is prohibited if its negation is obligatory

```
lemma kingdom-of-ends-1:
```

```
\begin{array}{l} \mathbf{shows} \models ((O\ \{A\}) \rightarrow (\square\ (P\ A))) \\ \mathbf{oops} \end{array}
```

- One interpretation of the categorical imperative is that something is obligatory only if it is permissible in every ideal world
- This formulation mirrors the kingdom of ends.
- This formulation is already a theorem of carmo and jones' DDL!.
- It can be shown using the O diamond rule, which just says that obligatory things must be possible.
- There are two possibilities: either the logic is already quite powerful OR this formulation is "empty".

```
\begin{array}{l} \mathbf{lemma} \ \textit{kingdom-of-ends-2:} \\ \mathbf{shows} \models ((\Box \ (P \ A)) \rightarrow (O \ \{A\})) \\ \mathbf{oops} \end{array}
```

- Notice also that ideally, this relationship does not hold in the reverse direction.
- Plenty of things are necessarily permissible (drinking water) but not obligatory.
- Very strange that this is a theorem in this logic.....
- That being said, Isabelle seems quite upset with this proof and is very slow to resconstruct it
- I am struggling to recreate this proof on paper

```
lemma permissible-to-ob:

shows \models ((P A) \rightarrow (O \{A\}))

oops
```

- Uh-oh.....this shouldn't be true...
- Not all permissable things are obligatory.....

```
lemma weaker-permissible-to-ob:
```

```
\mathbf{shows} \models ((\lozenge (P A)) \to O \{A\})
```

- Makes sense that this follows from the reverse kingdom of ends.
- Obligation and necessity/possibility are separated in this logic
- Both the dyadic obligation and necessity operator are world agnostic

```
{\bf lemma}\ contradictory-obligations:
```

```
shows \models (\neg ((O \{A\}) \land (O \{\neg A\})))
nitpick[user-axioms]
oops
```

- What is the cause of the above strangeness?
- This very intuitive theorem holds in my logic but not in BFP's
- It's clear that this theorem results in the strange results above.
- Conclusion: There is a bug in my embedding
- Nitpick found a counterexample for card i = 2:

Free variable: $A = (\lambda x.)(i_1 := False, i_2 := True)$

Sidebar: the above theorem is really intuitive - it seems like we wouldn't want contradictory things to be obligatory in any logic. But for some reason, not only is it not a theorem of Carmo and Jones' logic, it actually implies some strange conclusions, including that everything is either permissible or obligatory. It's not clear to me from a semantic perspective why this would be the case. In fact this theorem seems like a desirable property. Potential avenue for exploration

Did some debugging. What was the problem? A misplaced parentheses in the definition of ax5b that led to a term being on the wrong side of an implication. Computer Science:(

After the debugging, all of this is no longer true! On to the next attempt :) end

5 The Categorical Imperative

Simple Formulation of the Formula of Universal Law 5.1

This is my second attempt at formalizing the Formula of Universal Law

```
abbreviation ddlpermissable::t \Rightarrow t (P-)
  where (P A) \equiv (\neg (O \{\neg A\}))
```

- This operator represents permissibility
- Will be useful when discussing the categorical imperative
- Something is permissible if it is not prohibited
- Something is prohibited if its negation is obligatory

Let's consider a naive reading of the Formula of Universal Law (FUL). From the Groundwork, 'act only in accordance with that maxim through which you can at the same time will that it become a universal law'. What does this mean in DDL? One interpretation is if A is not necessarily permissible, then its negation is obligated.

```
axiomatization where
FUL-1: \models ((\neg(\Box (P A))) \rightarrow (O \{(\neg A)\}))
```

5.2**Basic Tests**

```
lemma True nitpick [satisfy,user-axioms,format=2] oops
— "Nitpick found a model for card i = 1:
Empty assignment"
— Nitpick tells us that the FUL is consistent
— "oops" after Nitpick does not mean Nitpick failed.
lemma something-is-obligatory:
 shows \forall w. \exists A. O \{A\} w
 nitpick [user-axioms]
 oops
— We might think that in every world we want something to be obligated.
— Sadly, Sledgehammer times out trying to prove this. Let's relax this
— "Nitpick found a counterexample for card i = 1:
Empty assignment"
— Nitpick to the rescue! The formula is in fact not valid.
lemma something-is-obligatory-2:
 shows \forall w. \exists A. O \{A\} w
 nitpick [user-axioms, falsify=false]
— "Nitpick found a model for card i = 1:
Skolem constant: A = (\lambda x.)(i_1 := True)"
```

```
— Nitpick tells us that the formula is consistent - it found a model where the formula is true.
```

— This means that our model is underspecified - this formula is neither valid nor inconsistent.

```
lemma something-is-obligatory-relaxed:
 shows \exists A w. O \{A\} w
 nitpick [user-axioms]
 oops
— "Nitpick found a counterexample for card i = 1:
Empty assignment"
— The relaxed version definitely isn't valid.
\mathbf{lemma}\ something\ is\ obligatory\ relaxed\ -2:
 shows \exists A w. O \{A\} w
 nitpick [user-axioms, falsify=false]
 oops
 - "Nitpick found a model for card i = 1:
Skolem constant: A = (\lambda x.)(i_1 := True)"
— Nitpick tells us that the formula is consistent - it found a model where the
formula is true.
— The model seems underspecified.
5.3
       Specifying the Model
Let's specify the model. What if we add something impermissible?
\mathbf{consts}\ M{::}t
abbreviation murder\text{-}wrong::bool where murder\text{-}wrong \equiv \models (O \{ \neg M \})
lemma something-is-obligatory-2:
 assumes murder-wrong
 shows \forall w. \exists A. O \{A\} w
 using assms by auto

    It works this time, but I think "murder wrong" might be too strong of an

assumption
abbreviation poss-murder-wrong::bool where poss-murder-wrong \equiv \models (\lozenge (O \{ \neg \}))
M\}))
lemma wrong-if-posibly-wrong:
 assumes poss-murder-wrong
 shows murder-wrong
 using assms by blast
— This lemma holds and uses a slightly weaker assumption. This also seems to
get closer to the "heart" of this naive interpretation. We really want to say that if
```

Let's try an even weaker assumption: Not everyone can lie.

something isn't necessarily obligated, it's not obligated anywhere."

typedecl person

```
consts lies::person \Rightarrow t
consts me::person
lemma breaking-promises:
  assumes \neg (\forall x. lie(x) cw) \land (lie(me) cw)
  shows (O \{ \neg (lie(me)) \}) cw
  nitpick [user-axioms]
 oops
— No proof found. Makes sense:
— This version of FUL simply universalizes across worlds (using the \square operator),
— But NOT across people, which is really what the most obvious reading of FUL
  - "Nitpick found a counterexample for card person = 2 and card i = 2:
Free variable: lie = (\lambda x.)(p_1 := (\lambda x.)(i_1 := \text{True}, i_2 := \text{False}), p_2 := (\lambda x.)(i_1 := \text{False})
False, i_2 := \text{False})"
lemma universalizability:
  assumes \models O \{(lie(me))\}
 shows \forall x. \models (O \{(lie(x))\})
 nitpick [user-axioms] oops
 - Nitpick found a counterexample for card person = 2 and card i = 2:
Free variable: lie = (\lambda x.)(p_1 := (\lambda x.)(i_1 := \text{False}, i_2 := \text{True}), p_2 := (\lambda x.)(i_1 := \text{False}, i_2 := \text{True})
False, i_2 := \text{False}) Skolem constant: x = p_2
— This lemma demonstrates the point even more clearly - we really want to think
that obligations are consistent across people, but because we don't have a notion
of agency, we don't have that property.
```

5.4 Consistent Sentences

The above section tested validity. We might also be interested in some weaker properties

Let's test whether certain sentences are consistent - can we find a model that makes them true?

```
lemma permissible:

fixes A

shows ((\neg (O \{A\})) \land (\neg (O \{\neg A\}))) w

nitpick [user-axioms, falsify=false] oops

— "Nitpick found a model for card i = 1:

Free variable: A = (\lambda x..)(i_1 := False)"

— Awesome! Permissible things are consistent
```

- Awesome! Permissible things are consistent clearly we've fixed the bug from categorical_imperative_1
- Note that apparently it's not clear @cite kitcher if Kant actually thought that permissibility was a coherent concept. Either way, I think permissibility is a pretty widely accepted ethical phenomenon.

```
lemma conflicting-obligations:
fixes A
shows (O \{A\} \land O \{ \neg A\}) w
```

```
nitpick [user-axioms, falsify=false] oops
  "Nitpick found a model for card i = 2:
Free variable: A = (\lambda x.)(i_1 := \text{False}, i_2 := \text{True})"
— Oh no! Nitpick found a model with conflicting obligations - that's bad!
5.5
       Metaethical Tests
\mathbf{lemma}\ FUL\text{-}alternate:
 \mathbf{shows} \models ((\lozenge (O \{ \neg A \})) \rightarrow (O \{ \neg A \}))
— One problem becomes obvious if we look at the definition of permissible
— Expanding the FUL gives us: \sim \square \sim O(\sim A) \longrightarrow O(\sim A)
— By modal duals we get that \diamond O(\sim A) \longrightarrow O(\sim A)
— This means that if something is possibly prohibited, it is in fact prohibited.
— I'm not convinced that this is a desirable property of an ethical theory.
lemma arbitrary-obligations:
 fixes A::t
 shows O\{A\} w
 nitpick [user-axioms=true] oops
   "Nitpick found a counterexample for card i = 1:
Free variable: A = (\lambda x.)(i_1 := False)"
— This is good! Shows us that any arbitrary term isn't obligatory.
{\bf lemma}\ removing\mbox{-}conflicting\mbox{-}obligations:
  assumes \forall A. \models (\neg (O \{A\} \land O \{\neg A\}))
 shows True
 nitpick [satisfy, user-axioms, format=2] oops
— "Nitpick found a model for card i = 1:
Empty assignment"
— We can disallow conflicting obligations and the system is still consistent - that's
good.
lemma implied-contradiction:
 fixes A::t
 fixes B::t
 assumes \models (\neg (A \land B))
 shows \models (\neg (O \{A\} \land O \{B\}))
 nitpick [user-axioms]
proof -
 have \models (\neg(\Diamond(A \land B)))
   by (simp add: assms)
 then have \models (\neg (O \{A \land B\})) by (smt\ carmojones\text{-}DDL\text{-}completeness.O\text{-}diamond)
 thus ?thesis oops
— [4] mentions that if two maxims imply a contradiction, they must not be willed.
— Above is a natural interpretation of this fact that we are, so far, unable to prove.
— "Nitpick found a counterexample for card i = 2:
Free variables: A = (\lambda x.)(i_1 := \text{True}, i_2 := \text{False}) B = (\lambda x.)(i_1 := \text{False}, i_2 := \text{False})
```

True)"

— This isn't actually a theorem of the logic as formed - clearly this is a problem.

```
lemma distribute-obligations-if:

assumes \models O \{A \land B\}

shows \models (O \{A\} \land O \{B\})

nitpick [user-axioms, falsify=true, verbose]

oops
```

- Nitpick can't find a countermodel for this theorem, and sledgehammer can't find a proof.
- Super strange. I wonder if this is similar to $\Box(A \land B)$ vs $\Box A \land \Box B$

$\mathbf{lemma}\ distribute\text{-}boxes:$

```
assumes \models (\Box(A \land B))
shows \models ((\Box A) \land (\Box B))
using assms by blast
```

— We really expect the O operator to be acting like the \square operator. It's like a modal necessity operator, like necessity across ideal worlds instead of actual worlds. Therefore, we'd expect theorems that hold of \square to also hold of O.

```
\mathbf{lemma}\ \textit{distribute-obligations-only} if:
```

```
assumes \models (O \{A\} \land O \{B\})
shows \models O \{A \land B\}
nitpick [user-axioms] oops
```

— "Nitpick found a counterexample for card i = 2:

Free variables: A = $(\lambda x.)(i_1 := \text{True}, i_2 := \text{False})$ B = $(\lambda x.)(i_1 := \text{False}, i_2 := \text{True})$ "

- If this was a theorem, then contradictory obligations would be ruled out pretty immediately.
- Note that all of this holds in CJ's original DDL as well, not just my modified version.
- We might imagine adding this equivalence to our system.

lemma ought-implies-can:

```
shows \forall A. \models (O \{A\} \rightarrow (\Diamond A)) using O-diamond by blast
```

— "ought implies can" is often attributed to Kant and is a pretty basic principle - you can't be obligated to do the impossible. I'm not surprised that our base logic has this as an axiom. It's often said to be the central motivation behind deontic logics.

 \mathbf{end}

```
theory kroy
imports carmojones-DDL
```

This theory will contain a formalization of the CI based on Moshe Kroy's partial formalization. [5]

6 Kroy's Formalization of the Categorical Imperative

6.1 The Substitution Operator

6.1.1 Definition

typedecl s — s is the type for a "subject", like the subject of a sentence — Intuitively, we need some notion of "x does action", which we can write as "x is the subject of the sentence 'does action"

```
type-synonym os = (s \Rightarrow t)
```

- An open sentence is a generalized version of Kroy's substitution operator [5] 196
- "does action" is an open sentence that can be instantiated with a subject
- "P sub (d/e)" can be written as "S(e)", where S(d) = P
- So the terms that we substitute into are actually instantiations of an open sentence, and substitution just requires re-instantiating the open sentence with a different subject

6.1.2 Abbreviations

```
abbreviation os\text{-}neg::os\Rightarrow os (¬-)
where (¬A) \equiv \lambda x. ¬(A(x))
abbreviation os\text{-}and::os\Rightarrow os\Rightarrow os (-\wedge-)
where (A \wedge B) \equiv \lambda x. ((A(x)) \wedge (B(x)))
abbreviation os\text{-}or::os\Rightarrow os\Rightarrow os (-\vee-)
where (A \vee B) \equiv \lambda x. ((A(x)) \vee (B(x)))
abbreviation os\text{-}ob::os\Rightarrow os (O{-})
where O{A} \equiv \lambda x. (O {A(x)})
```

— We could probably do without these abbreviations, but they will simplify the notiation a bit and unify it with Kroy's original paper.

```
abbreviation ddl-permissible::t\Rightarrow t\ (P\ \{-\}) where P\ \{A\} \equiv \neg\ (O\ \{\neg\ A\}) abbreviation os-permissible::os\Rightarrow os\ (P\ \{-\}) where P\ \{A\} \equiv \lambda x.\ P\ \{A(x)\}
```

— Carmo and Jones don't make much use of permissibility, but we will find it useful here.

6.2 Differences Between Kroy's Logic (Kr) and DDL

[5] uses a different logic than DDL. Let's see if the semantics that Kr requires hold in DDL

```
lemma permissible-semantic-faithful:
fixes A w
shows P \{A\} w \longrightarrow (\exists x. A(x))
nitpick[user-axioms] oops
```

— The most faithful interpretation of Kr is that if A is permissible in a context, then it must be true at some world in that context. Kr operates under the "deontic alternatives" view, summarized by Solt as "A proposition of the sort OA is true at the actual world w if and only if A is true at every deontic alternative world to t." Under this view, permissible propositions are obligated at some deontic alternative, but not at all of them.

DDL does not adopt a deontic alternatives view, which is why this proposition seems wildly counterintuitive in DDL. In DDL, the ob operator abstracts away the notion of deontic alternatives and completely determines obligations. Even if one belives that permissible statements should be true at some deontic alternative, it's not clear that permissible statements must be realized at some world, hence the failure of this lemma in DDL.

```
— Nitpick found a counterexample for card i = 1:
Free variable: A = (\lambda x..)(i_1 := False)
```

```
\mathbf{lemma}\ permissible\text{-}semantic\text{-}faithful\text{-}2\colon
```

```
fixes A w shows P {A} w \longrightarrow (\exists x. O \{A\} x) nitpick[user-axioms] oops

— Nitpick found a counterexample for card i = 1:
Free variable: A = (\lambda x...)(i_1 := False)
```

— This is the most clear lemma that we would expect to hold under the deontic alternatives view. The fact that it doesn't shows DDL is not a logic of deontic alternatives. There are pros and cons to this approach. The deontic alternatives view is quite simple to visualize and offers clear intuition. On the other hand, DDL's ob function can encode more complex relations than the deontic alternatives view, and can encode these in a more intuitive manner. The notion of a "deontically perfect alternative" is a squishy one, and an ob function more directly captures the idea of obligation.

```
lemma permissible-semantic-vacuous: fixes A w
```

```
shows P \{A\} \ w \longrightarrow (\exists x. \ ob(x)(A))
nitpick[user-axioms] oops
```

— Kr does not allow vasuously permissible statements -; if something is permissible it has to be obligated at some deontically perfect alternative. In DDL, this can be roughly translated as, if A is permissible, it is obligated in some context.

```
— Nitpick found a counterexample for card i = 1:
```

```
Free variable: A = (\lambda x.)(i_1 := False)
```

— In order to make this true, we'd have to require that everything is either oblig-

atory or prohibited somewhere. But as found in the buggy version of DDL, that breaks everything and destroys the notion of permissibility everywhere. I am going to allow for vacuous permissibility. If something breaks later in Kr, it may be because of this.

```
lemma permissible-ob: fixes A w shows O \{A\} w \longrightarrow P \{A\} w nitpick [user-axioms] oops

— Nitpick found a counterexample for card i=2: Free variable: A = (\lambda x...)(i_1 := \text{False}, i_2 := \text{True})

— This one is definitely problematic. Being permissible should be a precondition for being obligatory. In my eventual logic, I will need to add this as an axiom, because I can't see any ethical theory succeeding without this.
```

6.3 The Categorical Imperative

```
abbreviation FUL::bool where FUL \equiv \forall w \ A. \ ((\exists p. \ (P \{A\}p)w) \longrightarrow (\forall x. \ (P \{A\}x)w))
```

— This is Kroy's formalization of the FUL in DDL. Recall that the FUL says "act only in accordance with that maxim through which you can at the same time will that it become a universal law" [3] Kroy interprets this to mean that if an action A is permissible for some agent p, then it must be permissible for everyone. This formalizes the important moral intuition that the formula of universal law prohibits free-riding. No one is a moral exception

```
lemma FUL:
  shows FUL
 nitpick[user-axioms] oops
— Nitpick found a counterexample for card s = 2 and card i = 2:
Skolem constants: A = (\lambda x.)(s_1 := (\lambda x.)(i_1 := \text{True}, i_2 := \text{True}), s_2 := (\lambda x.)(i_1 := \text{True})
:= False, i_2 := False)) p = s_1 x = s_2 This formalization doesn't hold in DDL. Good
- this means that adding it as an axiom will change the logic.
\mathbf{lemma} \models (\neg (O \{ \neg \top \}))
 by (simp \ add: \ ax-5a)
lemma complete:
  shows (\models (((A \rightarrow \neg \top)))) \longrightarrow (\models (\neg (O \{A\})))
  have (\exists x. (\neg (\Diamond A)) x) \longrightarrow (\exists x. (\neg O \{A\}) x)
    by (simp\ add:\ ax-5a\ ax-5b)
  thus ?thesis
    by blast
qed
axiomatization where CI: CI and
possible: (\forall w. \neg \Diamond A \ w) \longrightarrow \models (O \{\neg A\})
```

— We really need a way to add negative obligations - to related the concept of

```
contradiction and obligation. This seems reasonable - if A is never possible at any world, then it's prohibited. and hmm: \models (\neg (O \{ \neg \top \}))
```

6.4 Tests

```
lemma True nitpick [satisfy,user-axioms] oops
Nitpick found a model for card s = 1 and card i = 1:
Empty assignment
The categorical imperative is consistent!
```

6.4.1 Specifying the Model

```
lemma breaking-promises:

fixes me::s

fixes lie::os

assumes \exists x. (\neg (\Diamond(lie(x))) cw)

shows \exists x. (\neg (O \{lie(x)\})) cw

by (metis \ assms \ ax-5a \ ax-5b)
```

lemma breaking-promises:

```
fixes me::s
fixes lie::os
assumes \exists x. (\neg (\Diamond(lie(x))) cw)
shows \exists x. (\neg (P \{lie(x)\})) cw
nitpick[user-axioms] oops
```

6.4.2 Metaethical Tests

6.4.3 Kroy's Tests

end

References

- [1] C. Benzmüller, A. Farjami, and X. Parent. Dyadic deontic logic in hol: Faithful embedding and meta-theoretical experiments. In M. Armgardt, H. C. Nordtveit Kvernenes, and S. Rahman, editors, New Developments in Legal Reasoning and Logic: From Ancient Law to Modern Legal Systems, volume 23 of Logic, Argumentation & Reasoning. Springer Nature Switzerland AG, 2021.
- [2] J. Carmo and A. Jones. Completeness and decidability results for a logic of contrary-to-duty conditionals. *J. Log. Comput.*, 23:585–626, 2013.
- [3] I. Kant. *Groundwork of the Metaphysics of Morals*. Cambridge University Press, Cambridge, 1785.

- [4] C. Korsgaard. Kant's Formula of Universal Law. *Pacific Philosophical Quarterly*, 66:24–47, 1985.
- [5] M. Kroy. A partial formalization of kant's categorical imperative. an application of deontic logic to classical moral philosophy. *Kant-Studien*, 67(1-4):192–209, 1976.